



MILATARI NEWSLETTER

Volume 2 Number 5

April 1983

Price \$1.00

**** Milwaukee Area Computer Users' Council Meets ****

A byproduct of the recent computer faire at Brookfield Square is the organization of a computer users council. The organization of this group is being spear-headed by our own Gary Nolan and Larry Leskovsek.

Representatives from several of the computer users groups in the Milwaukee area held their first organizational meeting on Tuesday, March 17th. Discussions centered on what the activities of this organization should be. The ideas which were expressed were:

- A newsletter to list clubs, resources and coming events.

- Organize and operate an annual compute faire.

- Act as clearing house for computer owners looking for clubs.

- Maintain a speakers pool, list of teachers, etc.

- Obtain price considerations from vendors.

- Inform clubs about legistration which affects micro-computer owners.

- Establish contacts with computer education groups.

- Represent the common interest of all groups to the public.

The council plans to meet monthly to continue the organization effort.

BALLOT-CENSUS Enclosed

If you are a paid member of Milatari you should have combination ballot and census included with your newsletter. The top half is your ballot for the adoption of the bylaws and election of officers. The bottom half contains a listing of your name and address. Please review the data and make necessary corrections so we may update our file. Use this form to elect the type of membership you want (individual, family or associate). If you are choosing the family membership, please include the names of all family members.

If you are not attending our next meeting, please complete the census form and mail it to:

MILATARI Newsletter
P.O. Box 1191
Waukesha, WI 53187

MILATARI * * * APRIL 1983

Milwaukee Area ATARI Users Group

This newsletter is written and printed by members of the Milwaukee Area ATARI Users Group (MILATARI), an association of individuals with a common interest in using and programming ATARI computers. MILATARI is not affiliated with the ATARI company, nor any other commercial organizations.

All articles are written and donated by the membership. Opinions expressed in this publication are those of the individual author and do not necessarily represent, nor reflect, the opinions of MILATARI nor those of any other commercial or non-commercial organizations. Any article appearing in this newsletter may be reproduced, providing credit is given to the author and to MILATARI.

Write MILATARI Newsletter at P.O. Box 1191, Waukesha, WI 53187.

MEMBERSHIP INFORMATION

Membership is open to individuals and families who are interested in using and programming ATARI computers. The membership includes the subscription to this newsletter and access to the user's library. The membership fee is \$15 per year for individual, \$20 for family and \$10 for associate. Contact Larry Leskovsek, Treas. at 547-0249 or write MILATARI, P.O. Box 1191, Waukesha, WI 53187 for more information.

MEETING INFORMATION

MILATARI meetings are held once monthly. This month the meeting will be held at the Armbruster School, 7000 Greenway, Greendale, WI. The meeting is held in the multi-purpose room. BASIC classes begin at 2:00 P.M. Technical sessions are also held at 2:00 P.M. The business session begins at 3:00 P.M. followed by demonstrations. The library will be open before and after the business meeting.

MILATARI Officers:

President	Gary Nolan 353-9716
Vice-president	Nick Liberski 786-8434
Secretary/ Treasurer	Larry Leskovsek 547-0249
Education	Linda Scott 466-2314
Cassette Librarian	Ron Friedel 354-1717
Disk Librarian	Steve Booth 367-8739
Publications Librarian	Karl Buschhaus 774-2576
Newsletter Editor	David Frazer 542-7242

- - - - -

Technical support Group:

The following members have indicated a willingness to assist MILATARI members.

William Lawrence	1-968-3082 Programming
Don Wilcox	228-1650 Programming
Erik Hanson	252-3146 Prog/Tech
Gary Nolan	353-9716 Prog/Tech
David Frazer	542-7242 Prog/Tech
Steve Booth	367-8739 Programming

- - - - -

MILATARI Bullentin Board:

The MILATARI Users Group maintains a 24 hr bulletin board service. Messages may be posted and read and public domain programs uploaded and downloaded. The system operates at 300 baud. The phone number is 352-2772.

PRESIDENT'S RAM

by Gary Nolan

A chicken in every pot.
A computer in every house.
And a hard disk for every computer!

Would you believe a firm handshake, a pat on the back, and a hearty hi-ho election????

Well, the officers you elect at the April meeting can't promise you the first three. And maybe only two of the second group. But what they can and will promise, is to do their best in the coming year. And the next twelve months holds the promise of being a very good year. So be there on April 16th to elect the persons of your choice to run MILATARI for the next twelve months. Nominations will be open up to meeting time, so there's still time to put someone up for office. In addition to voting for officers we will be adopting our by-laws. Now you have twice the incentive to be there.

KIDS SAY THE DARNDDEST THINGS:

The last couple of months we've been talking about ATARI's computer marketing(?) department. A clue to the origin of their thinking can be found in this month's PERSONAL COMPUTING. In it they do an interview with John Cavalier, president of the home computer division. He came to ATARI from American Can Co. where he was V-P of the Dixie Cup division. Now there's a real high-tech product for you. After reading this article you begin to see why their thinking is screwed up. Some gems from the interview follow.

When asked what impact IBM's rumored "home" computer would have on the market John answered, "Where we have an advantage is that we understand the consumer better than IBM does."

Q. The 1200XL has an RGB output. Will you offer an RGB monitor?

A. You'll see a lot of interesting things on our computers. (I'd sure like to know where the RGB output is on the 1200. 'Cause I haven't found it! But I sure like how he avoided answering the the question. Maybe he doesn't know what RGB means. After all Dixie cups don't have but one input/output port.)

Q. (On mass data storage) What will be used in the home?

A. I believe you'll see lots of disk storage in the home. That leads to products with built in disk drives that allow you to manipulate disks very quickly. I'm not saying we're going to do that, but it makes a lot of sense. (Excuse me.....THUD-THUD-THUD.....one of these days I'll have to pad that wall.)

These are just some of the answers that either strike fear into your heart or make you want to laugh. He does make some valid points though. On the promise of word processing in the home, the value of entry level machines (Sinclair?) and his definition of a "home" computer. And the most important one, that the (computer) INDUSTRY has to make sure people know what to do with the machines when they get them home.....AHMEN!

FYI:

ATARI is having trouble on other fronts. First they alienated the computer dealers by going to a mass merchant selling plan, now they're getting discount stores mad by insisting on exclusive distributor set-ups. Several chains would rather drop the line than work with 10/15 different distributors. This and a lack of information on upcoming products is causing discord among dealers. ATARI is also having little success in trying to keep dealers from discounting the 1200. ATARI has "suggested" a

\$749 bottom price to dealers. Some, in order to sell a reasonable quantity, have had to go well below that price. Oh well, FORD survived the EDSEL. APPLE survived the III, and we all survived NIXON. So maybe ATARI can survive the 1200XL. Look at it this way, as a replacement for the 400 (not counting the price difference) it's a good machine and it gives us something to talk about when things get slow.

While we're on the subject of the 400. ATARI will be running a \$50 special rebate starting April 15th and running until Sept. 30th. You can buy one from TARGET this week (April 10/16) for \$140+tax. ATARI will honor the rebate coupon even though it's not the 15th yet.

Latest rumors:

ATARI will announce NEW hardware at the summer CES in June.

Maybe it will be ATARI-TEL (or Project FALCON) a computerized telephone system from ATARI. What it sounds like is either a computer with a built in modem or a telephone with a built in computer and appliance controller (ala the BSR type?). They're scheduled for 1984 release but they hope to have it ready by late this year.

ATARI is also rumored to be ready to bring out a new line of computers, the XL series, of which the 1200 was the first. As they say, "Don't believe it 'till you see it, don't trust it 'till you've got it in your hands."

Books, Magazines, Newspapers:

As mentioned at last months meeting, we are going to expand the publications library. Added this month and available for check-out will be the following:

- Computes First Book of ATARI
- Computes Second Book of ATARI
- Computes First Book of ATARI Graphics
- Mapping The ATARI
- The ATARI Assembler

We're trying to locate another copy of the Tech Manuals around here but have not found any. But we do plan to add another copy to the pile. As this is one of the more popular items we may have to impose a penalty on those who keep it for more than one month! If you know of a book that's not in the library that you think should be there, let us know about it. If there is enough interest in any book we can put together a bulk purchase and reduce the cost a little.

Computer Shopper a monthly newspaper is running a special promotion thru the end of May of 50% off the regular subscription price for user group members. Normal rate is \$10, so you get it for \$5. For those of you who haven't seen this newspaper there will be one for inspection at the meeting. I will have a subscription form available for those who wish to sign up. CS is made up of about 90% ads. Some for used equipment a lot for new.

ANTIC has gone to a monthly publishing schedule. The quality of this magazine continues to improve and is a must for ATARI owners.

PRICE WARS (or, How low can you go Darth Vader???)

It usually happens around Christmas time, but what the hay! Everybody is cutting prices. In the same ad that TARGET had for the 400 that was mentioned earlier they had a VIC-20 for \$90. I've seen a COMMODORE 64 priced at \$389. The TI 99/4A is selling for \$150, and going into effect from June 1st till Jan.31,1984 will be a \$50 rebate. Leapin' Lizards Sandy, a real computer with a real keyboard for ONE HUNDRED DOLLARS! Getting down to the TIMEX TS1000 range, right? Wrong! TIMEX dropped that price with rebate to around 45/50 dollars. And these are LOCAL prices, NOT mail order! What's next? Packages! All sorts of combo's. Hardware, software and both.

MILATARI * * * APRIL 1983

Look at KAYPRO, MORROW DESIGNS, OSBORNE and other "personal" types. The same thing will probably happen to the "home" field. If you can remember back that far, the 800 was originally sold with the cassette recorder, BASIC cartridge, System master cart. and Invitation to Programming cassette. And several manufacturers have announced similar ones. Where will it all end? Who knows! One other point that John Cavalier made in the aforementioned interview was that those who don't learn from history are damned to repeat it. There used to be a lot of companies in the calculator and digital watch business. And in their attempt to gain market share they forgot about bottom line profit. Those that forgot are not in the business today. I'd like to add to that, that some people went out of the business because their products were overpriced. In the short term the consumer reaps the rewards in the form of lower prices. But over the long haul somebody comes up on the short end. It's those people who buy one of the systems whose maker disappears after a year. "Let the buyer beware"

IS ANYBODY OUT THERE LISTENING??:

Well ATARI seems to be. To some kids anyway. It seems that ATARI has what it calls the ATARI Youth Advisory Board. Made up of twenty young people from around the country, they recently took part in a three day brainstorming session. They offered opinions on products, marketing, advertising and promotions. Those involved ranged in age from 13 to 17 and all can program in at least one language. Not all were ATARI owners or users. Although they were not paid for the session they did receive some compensation. Like 1200XL's, software and a chance to tell ATARI what they're doing right AND wrong. Ah! To be young again. As ATARI's VP of public affairs said, "ATARI listens to youth. They are our future executive and users." So what are their current users, chopped liver? I wonder what we would have to do to have them listen to us? As the Missouri mule trainer said after he wacked his pupil on the head with a big stick, "First you have to get their attention." MMMMM I've got the bat, anyone have a spare plane ticket to San Jose?

NEW EQUIPMENT:

There are a lot of new printers on the market today. But one of the more interesting ones is the new OLIVETTI PR2300 ink jet printer. It runs at 120 cps with three type sizes, three underline styles and four characters styles. Uses single sheet or roll paper to 8 1/2" wide and sprocket fed to 9 1/2" wide. Features include out of paper and out of ink detectors, 1K buffer and a CENTRONICS parallel interface. The printer comes with paper roll sprockets, dust cover and 4 ink jet cartridges (each good for about 150,000 char.). Standard 90 day parts/labor warranty. In this area it's available from;

QUILL Corp.

100 S. Schelter Road

Lincolnshire, IL 60069

Another nice thing about it is that it's on sale until the end of the month. The price? How does \$449.88 sound? Want one, or more info? Call 1-312-634-4800 to order, 1-312-634-4850 for more information.

MISC INFO:

It seems like we've done this before but we'll do it again. Those interested in buying disks can do so at the meeting. Regular price \$19 a box of ten. We may have some cassettes for sale also.

One new book that will be used as a reference book, (until we buy another one) is The Book of ATARI Software 1983. It breaks programs down into four categories. Games & entertainment, business, education and utility programs. Although each category has its own rating criteria (12 for

MILATARI * * * APRIL 1983

games, 8 for business, etc), every criteria is rated A (superior) to F (unacceptable). Of course not every program is listed, but it does cover more than 400 programs for the 800/400 computers, 2600 & 5200 games. I also reviews some hardware and books. Cost is \$20 but well worth it.

THE END:


Well boys and girls, we've come to the end of another column and it's time to say good-bye. So see you on the 16th. BYE.....

(from the TWIN CITY ATARI GROUP NEWSLETTER)


SOUND BY TRB

As a lot of you saw at the last meeting, the ATARI can produce some pretty impressive sounds when channelled through a stereo amplifier and a reverb unit. For those of you that missed the meeting, just imagine what STAR RAIDERS sounds like on your TV speakers, then imagine what the sound would be like given the full range (20 hz bass to 20000 hz treble) of a stereo system, and finally imagine what the sound would be like if you could add depth (deep cavern effect), repeats (echoing phasor shots), and delay (stereo effect) to one channel of the stereos' sound. If you still can't imagine what the sound is like, go down to ON-LINE computers and look at their set up (they have small speakers, but you will get the idea). I would like to thank Mike Davis (a TAIG member) for recommending this set up to me (he is heavily into music and is always trying things out).

Now for the fun part...what does it cost to configure your ATARI for super sound? Very little if you have a stereo system. I decided not to take over our family stereo (my wife would kill me, since I already hog her TV), so I went out and purchased an inexpensive amplifier (Realistic SA-102: \$60.00), a pair of speakers (Realistic 8"/2": \$40.00 each), 25 feet of 2 conductor PVC insulated telephone cable (\$2.50), one 5-pin Audio/Video Plug (#274-003: \$1.49), two RCA type PHONO PLUGS (#274-339: \$1.39 ea), some speaker cables to connect up the components (just used RCA plugs soldered to speaker wire), and finally the heart of the system, the reverb unit (Realistic Electronic Reverb: \$40.00). All of these items may be purchased from Radio Shack. The total cost of everything but the amplifier and speakers was around \$55.00. The following diagram indicates how everything should be hooked up.



Following these directions to Armbruster School



DATA FILE PROCESSING

Storing data on the ATARI 410 (tm) Program Recorder and the ATARI 810(tm) Disk Drive

** Cassette tutorials appeared in the March issue.

- 1) Storing Data on Cassette
- 2) A Simple Data File on Cassette
- 3) An Example of Cassette I/O

** Disk tutorials

- 4) Storing Data on Disk
- 5) Example of Disk I/O: Disk Mailing List

** Random access tutorial will appear in the May issue.

- 6) Random Access

Information provided by:

ATARI INC.
CONSUMER PRODUCT SERVICE
PRODUCT SUPPORT GROUP

DEMOPAC #2 (con't from March issue)

DATA FILE PROCESSING
Storing Data on Disk
WBB 4/82

The ATARI 810 Disk Drive stores data on 5-1/4" floppy diskettes. A diskette is formatted into 40 concentric tracks, each with 18 sectors giving a total of 720 sectors. Each sector is 128 bytes long (single density). Therefore, the total storage capacity on each diskette is 92,160 bytes.

The 810 Disk Drive uses a *boot file* to control the power-up initialization procedure. This usually means that the Disk Operating System File Management System (DOS SYS) is loaded into RAM. The DOS FMS is responsible for allocation the available sectors on a disk as needed for file storage. The boot file uses 3 of the 720 sectors.

DOS maintains a *directory* of the sectors that have been created on the disk up to a maximum of 64 files. The files are located in the directory by a unique name and they are identified by their sequential position in the directory numbered 0-63. The directory takes 8 of the 720 sectors.

DOS maintains a *bit map* of the sectors that have been allocated for file storage. When DOS needs to allocate a sector to a file, it uses the lowest numbered free sector as indicated in the bit map. This is commonly referred to as *random access*. The bit map takes 1 of the 720 sectors.

DOS uses 3 of the 128 bytes in each sector to identify the file it belongs to and the next sequential sector in the file. Therefore, the file storage space available to the user with DOS 2.0S is 707 sectors x 125 bytes = 88,375 bytes.

OPEN

DOS maintains a *pointer* for each file opened. The pointer is the location in the file the next I/O command will access. The OPEN command establishes a channel from the token file in RAM to the 810 Disk Drive. There are eight channels in the system numbered 0-7. The OS uses 0,6,7 at various times so you should use 1-5. There are four modes for the OPEN command.

OPEN #1,4,0,"D:filename" opens a file in read mode. DOS locates FILENAME in the directory and positions the pointer at the first byte of the file. INPUT #1 or GET #1 are the only legal commands.

OPEN #1,8,0,"D:FILENAME" always opens a new file in WRITE mode. DOS first searches the directory for FILENAME and, if it exists, deletes it. DOS then creates FILENAME in the directory, allocates a free sector as the first sector of the file and positions the pointer at the first byte of that sector. PRINT#1 or PUT#1 are the only legal commands.

OPEN #1,9,0,"D:FILENAME" opens a file in APPEND mode. DOS first locates FILENAME in the directory. It then allocates a free sector appended to the end of the file and positions the pointer at the first byte of that sector. Mode 9 is write only, so PRINT#1 or PUT#1 are the only legal commands.

OPEN #1,12,0,"D:FILENAME" opens a file in READ/WRITE mode. DOS locates FILENAME in the directory and positions the pointer at the first byte of the file. PRINT#1, PUT#1, INPUT#1, and GET#1 are all legal commands in this mode. However, the output commands write over the values that are currently in the file so the user should be careful to replace the exact number of bytes when updating an existing field. It is not possible to append new data to a file in this mode. Mode 12 can also be thought of as RANDOM ACCESS mode because it is the only mode that supports the commands NOTE#1 and POINT#1.

CLOSE

The CLOSE command frees the channel from the program to the device. It is always a good idea in the interest of good programming technique to CLOSE every file that is opened in a program. However, it is imperative that a file opened in update modes 8 or 9 be closed. If not, it is probable that the bit map or forward pointers will not be updated correctly. The results is an error 164, requiring the disk to be reformatted.

PRINT/PUT

The output commands, PRINT#n and PUT#n, transfer data from the token file to the buffer for the disk file. When the buffer fills up with 125 bytes, DOS writes the buffer to the file, allocates another free sector and clears the buffer. Two types of I/O can be used to write data to a file; character I/O or record I/O.

Character I/O means that you write data one byte at a time with none of the values interpreted as control characters. The statement PUT#n,X transfers one ATASCII byte to the disk file.

Record I/O means that you write data one field at a time with the End of Line (EOL) character (ATASCII 155) used to delimit the end of the fields. The EOL character is automatically generated by the PRINT statement. Therefore, to avoid having to put in your own delimiters, simply transfer each field on a separate PRINT statement. The PRINT statement should

include a semicolon and not a comma (PRINT #1;A\$). A comma is interpreted as a tab, causing 10 blank spaces to be inserted in front of your data. The size of the field should be limited to less than 119 bytes to avoid using reserved memory space on page 6 of RAM.

INPUT/GET

The input commands, INPUT #1 and GET #1, transfer data from the disk file to the token file. The data should be read from the file in the same fashion that it was written, character or record I/O.

Character I/O reads one byte at a time with none of the values being interpreted as control codes. The statement GET #1,X transfers one ATASCII byte from the disk file to the variable X.

Record I/O reads one field at a time with EOL (ATASCII 155) used to delimit the end of each field (INPUT #1,A\$). Many fields can be transferred with each INPUT statement (INPUT #1,A\$,B\$,C\$).

There are three ways to read all that data from the file and exit without an error. If you know how many fields were written, you can simply read that number of fields. If the number of fields varies, you can write a special value at the end of the file and check for this value after each input. If you don't know what's in the file, you should use the TRAP command. When the end-of-file error 136 occurs, the TRAP will send you to an error routine. The routine should check that location 195 (error status) does contain a 136, and then CLOSE the file.

NOTE/POINT

NOTE#1,S,B stores the current sector and byte location of the pointer in the variable S and B. Conversely, POINT#1,S,B moves the pointer directly to the sector and byte specified in the variables. These commands are used together in mode 12 to provide the user with random access to the disk file. In general the procedure is NOTE#1, INPUT#1, POINT#1, and PRINT#1. This should only be done with fixed length records. If you update a 4-byte field with a 3-byte field, an extra EOL is added to the file and the number of fields is incorrectly incremented by 1. If you update a 4-byte field with a 5-byte field, the next sequential field in the file is overwritten. Care should be taken to replace the exact number of bytes when updating an existing field.

DATA FILE PROCESSING

An Example of DISK I/O
WBB/JB 3/82

The following set of programs sets up and maintains a simple mailing list using the 810 Disk Drive. The programs provide an example of one method of storing data in data files. The first program sets up the file by adding records. The second updates the information in the file. The third prints out the contents of the file.

The key concepts illustrated are opening a data file with the OPEN statement, and writing to that file using the PRINT#1; statement. In this simple example, only one variable is written at a time, so no extra data separators are necessary.

A temporary file is used to keep the updated information. When the update

MILATARI * * * APRIL 1983

is complete, the temporary file is renamed, and the old file becomes the temporary file. Some error trapping is provided.

```

1 REM DISKADD
2 REM WBB/JB 4/82
3 REM --use this program to create a file, or to add new records--
4 REM the program creates a temporary file, adds records to it,
5 REM then deletes the permanent file and renames the temporary file
6 REM so that it becomes the permanent file.
7 REM *****
10 PRINT CHR$(125);REM clear screen
20 PRINT "THIS PROGRAM ADDS RECORDS"
30 PRINT "FOR NEW CUSTOMERS.":PRINT
40 PRINT "INSERT THE PROPER DISKETTE"
50 PRINT "PRESS 'START' TO CONTINUE..."
60 IF PEEK(53279)<>6 THEN 60:REM wait for start key
65 REM --set up variables and filenames--
70 DIM ID$(9),NAME$(24),ADDR$(24),CITY$(16),STATE$(2),ZIP$(5),PHONE$(12)
80 DIM FILE1$(16),FILE2$(16)
90 FILE1$="D:CUSTOMER.DAT":FILE2$="CUSTOMER.TMP"
95 REM --open the files--
100 CLOSE #1:CLOSE #2:REM close any currently open files
110 TRAP 200:OPEN #1,4,0,FILE1$:TRAP 40000:REM check for 'no file found'
error
120 OPEN #2,8,0,FILE2$:REM if no error, open write file
130 GOTO 300:REM skip over error routine
180 REM *****
190 REM --error routine--
195 REM if the file is not on this disk then create one, or replace disk
200 PRINT CHR$(253);"CUSTOMER FILE NOT ON THIS DISK,":PRINT :REM sound
buzzer
210 PRINT "PRESS 'START' TO TRY ANOTHER DISK-"
220 PRINT "PRESS 'SELECT' TO CREATE ON THIS DISK-"
230 IF PEEK(53279)=6 THEN 100:REM if start is pressed, try again
240 IF PEEK(53279)<>5 THEN 230:REM check for select
250 CLOSE #1:OPEN #1,8,0,FILE1$:REM if creating, open write file
260 PRINT #1;"ENDOFFILE":REM write file with no records
270 CLOSE #1:GOTO 100:REM now that there is a file, go try again
280 REM *****
290 REM --transfer existing records to new file--
300 INPUT #1,ID$:REM get record number
305 IF ID$="ENDOFFILE" THEN 400:REM last record, go to add-record routine
310 PRINT #2;ID$:REM transfer number to new file
320 PRINT "TRANSFERRING TO TEMP FILE...";ID$
330 INPUT #1,NAME$:PRINT #2;NAME$
340 INPUT #1,ADDR$:PRINT #2;ADDR$
350 INPUT #1,CITY$:PRINT #2;CITY$
360 INPUT #1,STATE$:PRINT #2;STATE$
370 INPUT #1,ZIP$:PRINT #2;ZIP$
380 INPUT #1,PHONE$:PRINT #2;PHONE$
390 GOTO 300:REM get next record
395 REM *****
396 REM --add new records to file--
400 PRINT CHR$(125);"SPECIFY RECORD TO ADD";CHR$(29):REM (move cursor down)
410 PRINT "ID NUMBER OR 'END'...":INPUT ID$:IF ID$="END" THEN 600
420 PRINT "NAME...":INPUT NAME$
425 PRINT "ADDRESS...":INPUT ADDR$
430 PRINT "CITY...":INPUT CITY$

```


MILATARI * * * APRIL 1983

```

435 PRINT "STATE...";:INPUT STATE$
440 PRINT "ZIP...";:INPUT ZIP$
45 PRINT "PHONE...";:INPUT PHONE$
450 PRINT :PRINT "PRESS 'SELECT' TO ADD RECORD..."
460 PRINT "PRESS 'OPTION' TO RE-ENTER..."
470 IF PEEK(53279)=3 THEN 400:REM option is pressed, re-enter record
480 IF PEEK(53279)<>5 THEN 470:REM check for select key
485 REM *****
495 REM --write new record to temporary file--
500 PRINT #2;ID$
510 PRINT #2;NAME$
520 PRINT #2;ADDR$
530 PRINT #2;CITY$
540 PRINT #2;STATE$
550 PRINT #2;ZIP$
560 PRINT #2;PHONE$
570 GOTO 400:REM go get more new records
580 REM *****
590 REM --closing routine--
600 PRINT #2;"ENDOFFILE":REM write end-of-file marker
610 CLOSE #1:CLOSE #2
620 PRINT CHR$(125);"DELETING OLD FILE..."
630 XIO 33,#1,0,0,FILE1$:REM delete old file
640 PRINT :PRINT "REM>NAMING NEW FILE..."
650 XIO 32,#1,0,0,"D:CUSTOMER.TMP,CUSTOMER.DAT"
660 PRINT :PRINT "--END OF PROGRAM--"
670 END

```

```

1 REM DISK UPDATE
2 REM WBB/JB 3/82
3 REM --use this program to change or delete existing records in the file--
4 REM read records from permanent file, update temporary file,
5 REM then delete old file and rename new one to be new permanent file.
6 REM *****
10 PRINT CHR$(125):REM clear screen
20 PRINT "THIS PROGRAM CHANGES OR DELETES"
30 PRINT "EXISTING RECORDS IN THE DATA FILE.":PRINT
40 PRINT "PRESS 'START' TO CONTINUE...":PRINT
50 IF PEEK(53279)<>6 THEN 60:REM wait for start key
60 REM --set up variables and filenames--
70 DIM ID$(9),NAME$(24),ADDR$(24),CITY$(16),STATE$(2),ZIP$(5),PHONE$(12)
80 DIM FILE1$(16),FILE2$(16)
90 FILE1$="D:CUSTOMER.DAT":FILE2$="CUSTOMER.TMP"
95 REM --open the files--
100 CLOSE #1:CLOSE #2:REM close any currently open files
110 TRAP 200:OPEN #1,4,0,FILE1$:TRAP 40000:REM check for 'no file found'
error
120 OPEN #2,8,0,FILE2$:REM if no error, open write file
130 GOTO 300:REM skip over error routine
180 REM *****
190 REM --error routine--
195 REM if the file is not on this diskette, try another one
200 PRINT CHR$(253);"CUSTOMER FILE NOT ON THIS DISK,":PRINT :REM sound
juzzer
210 PRINT "PRESS 'START' TO TRY ANOTHER DISK-"
240 IF PEEK(53279)<>6 THEN 240:REM check for start key
250 GOTO 100:REM try again
280 REM *****

```


MILATARI * * * APRIL 1983

```

290 REM --read a record from the old file--
300 INPUT #1,ID$:REM get record number
305 IF ID$="ENDOFFILE" THEN 600:REM last record, go to add-record routine
310 INPUT #1,NAME$,ADDR$,CITY$,STATE$,ZIP$,PHONE$:REM read rest of record
315 REM -- display the record --
320 PRINT CHR$(125);"DATA IN OLD FILE";CHR$(29):REM (move cursor down)
330 PRINT "ID",ID$
335 PRINT "NAME",NAME$
340 PRINT "ADDRESS",ADDR$
345 PRINT "CITY",CITY$
350 PRINT "STATE",STATE$
355 PRINT "ZIP",ZIP$
360 PRINT "PHONE",PHONE$
370 PRINT :PRINT "PRESS 'OPTION' TO MODIFY RECORD-"
375 PRINT "PRESS 'SELECT' TO KEEP RECORD AS IS-"
380 PRINT "PRESS 'START' TO DELETE RECORD-"
390 IF PEEK(53279)=6 THEN 300:REM get another record, don't save this one
391 IF PEEK(53279)=5 THEN 500:REM add this record to the new file
392 IF PEEK(53279)<>3 THEN 390:REM check for option key
395 REM *****
397 REM -- modify data in record --
400 PRINT :PRINT "ENTER NEW DATA FOR RECORD":PRINT
410 PRINT "ID NUMBER...":INPUT ID$
420 PRINT "NAME...":INPUT NAME$
425 PRINT "ADDRESS...":INPUT ADDR$
430 PRINT "CITY...":INPUT CITY$
435 PRINT "STATE...":INPUT STATE$
440 PRINT "ZIP...":INPUT ZIP$
445 PRINT "PHONE...":INPUT PHONE$
450 PRINT :PRINT "PRESS 'SELECT' TO ADD RECORD..."
460 PRINT "PRESS 'OPTION' TO RE-ENTER..."
470 IF PEEK(53279)=3 THEN 400:REM option is pressed, re-enter record
480 IF PEEK(53279)<>5 THEN 470:REM check for select key
485 REM *****
495 REM --write new record to temporary file--
500 PRINT #2;ID$
510 PRINT #2;NAME$
520 PRINT #2;ADDR$
530 PRINT #2;CITY$
540 PRINT #2;STATE$
550 PRINT #2;ZIP$
560 PRINT #2;PHONE$
570 GOTO 400:REM go get more new records
580 REM *****
590 REM --closing routine--
600 PRINT #2;"ENDOFFILE":REM write end-of-file marker
610 CLOSE #1:CLOSE #2
620 PRINT CHR$(125);"DELETING OLD FILE..."
630 XIO 33,#1,0,0,FILE1$:REM delete old file
640 PRINT :PRINT "REM>NAMING NEW FILE..."
650 XIO 32,#1,0,0,"D:CUSTOMER.TMP,CUSTOMER.DAT"
660 PRINT :PRINT "--END OF PROGRAM--"
670 END

```

```

1 REM DISK PRINT
2 REM WBB/JB 3/82
3 REM --use this program to print out the customer list from the file--
6 REM *****

```


MILITARI * * * APRIL 1983

```

10 PRINT CHR$(125):REM clear screen
20 PRINT "THIS PROGRAM PRINTS ALL RECORDS"
 0 PRINT "FROM THE DATA FILE ON A PRINTER.":PRINT
40 PRINT "PRESS 'START' TO CONTINUE...":PRINT
50 IF PEEK(53279)<>6 THEN 50:REM wait for start key
60 REM --set up variables and filenames--
70 DIM ID$(9),NAME$(24),ADDR$(24),CITY$(16),STATE$(2),ZIP$(5),PHONE$(12)
80 DIM FILE1$(16),FILE2$(16)
90 FILE1$="D:CUSTOMER.DAT":FILE2$="P:"
95 REM --open the files--
100 CLOSE #1:CLOSE #2:REM close any currently open files
110 TRAP 200:OPEN #1,4,0,FILE1$:TRAP 40000:REM check for 'no file found'
error
120 TRAP 250:OPEN #2,8,0,FILE2$:TRAP 40000:REM trap printer-not-ready error
130 GOTO 300:REM skip over error routine
180 REM *****
190 REM --error routine--
200 PRINT CHR$(253);"CUSTOMER FILE NOT ON THIS DISK,":PRINT :REM sound
buzzer
210 PRINT "PRESS 'START' TO TRY ANOTHER DISK-"
220 IF PEEK(53279)<>6 THEN 220:REM check for start key
230 GOTO 100:REM try again
235 REM *****
240 REM --error routine, printer not ready--
250 PRINT CHR$(253);"PRINTER NOT READY,":REM sound buzzer
260 PRINT "PRESS 'START' TO TRY AGAIN...":PRINT
270 IF PEEK(53279)<>6 THEN 270:REM wait for start key
280 GOTO 100:REM try again
285 REM *****
290 REM REM read a record form the disk file --
300 INPUT #1,ID$:REM get record number
305 IF ID$="ENDOFFILE" THEN 500:REM last record, go to add-record routine
310 INPUT #1,NAME$,ADDR$,CITY$,STATE$,ZIP$,PHONE$:REM read rest of record
315 REM -- display the record --
320 PRINT CHR$(125);"DATA IN OLD FILE";CHR$(29):REM (move cursor down)
330 PRINT "ID",ID$
335 PRINT "NAME",NAME$
340 PRINT "ADDRESS",ADDR$
345 PRINT "CITY",CITY$
350 PRINT "STATE",STATE$
355 PRINT "ZIP",ZIP$
360 PRINT "PHONE",PHONE$
370 PRINT "PRESS 'SELECT' TO PRINT RECORD..."
380 PRINT "PRESS 'START' TO READ NEXT RECORD..."
390 IF PEEK(53279)=6 THEN 300:REM get another record
391 IF PEEK(53279)<>5 THEN 390:REM check for select key
395 REM *****
397 REM -- print out record on printer --
400 PRINT #2;ID$
410 PRINT #2;NAME$
420 PRINT #2;ADDR$
430 PRINT #2;CITY$
440 PRINT #2;STATE$
450 PRINT #2;ZIP$
460 PRINT #2;PHONE$
470 GOTO 390:REM go wait for ok to read new record
480 REM *****
495 REM -- closing routine --

```



```

500 CLOSE #1:CLOSE #2
510 PRINT CHR$(125):REM clear screen
520 PRINT "-- END OF PROGRAM --"
530 END

```

* * * * *

ANIMATION WITH THE JOYSTICK

by Ed Gombert

West Valley Atari User's Group Newsletter

One of the first things I discovered while playing around with my computer was how simple it was to use joysticks in a program. They are really nothing more than IF-THEN statements. In other words, "IF" the joystick is in this position "THEN" do this. To show how this works I have written a program that uses the joystick to animate a three dot "caterpillar" similar to the centipede game. As simple as the program is I will describe it briefly.

Line 10 starts out by clearing the screen. POKE 752,1 gets rid of the cursor. POKE 82,0 sets the left hand margin at 0 instead of 2 where it usually is. The three remaining POKES set the color for the background, border and characters.

Line 75 gives each section of the caterpillar the value of the section just in front of it. In line 90 thru 160 the first dot is the last section to be given a new value depending on which direction the joystick is pointing. STICK(0) is the first joystick port and the following number indicates the stick position as shown in the diagram below.

Line 170 thru 200 will send each section of the caterpillar to the opposite edge of the screen as they go off the sides, top or bottom. And of course line 210 sends the program back to line 30 to reposition the caterpillar based on the new information from the joystick.

CATERPILLER PROGRAM

```

10 ? "(clear)":POKE 752,1:POKE 82,0:POKE 709,12:POKE 710,20:POKE 712,18
20 A=3:B=0:C=2:D=0:E=1:F=0:G=0:H=0:I=0:J=0
30 POSITION A,B:?" "
40 POSITION C,D:?" "
50 POSITION E,F:?" "
60 POSITION G,H:?" "
70 POSITION I,J:?" "
75 I=G:J=H:G=E:H=F:E=C:F=D:C=A:D=B
80 IF STICK(0)=15 THEN 80
90 IF STICK(0)=7 THEN A=A+1
100 IF STICK(0)=13 THEN B=B+1
110 IF STICK(0)=5 THEN B=B+1:A=A+1
120 IF STICK(0)=9 THEN B=B+1:A=A-1
130 IF STICK(0)=11 THEN A=A-1
140 IF STICK(0)=14 THEN B=B-1
150 IF STICK(0)=6 THEN B=B-1:A=A-1
160 IF STICK(0)=10 THEN B=B-1:A=A-1
170 IF A>38 THEN A=1
180 IF B>22 THEN B=0
190 IF B<0 THEN B=22
210 GOTO 30

```

